# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=dnsrecon
pkgver=1.5.1
_pyver=3.13
pkgrel=1
epoch=2
groups=('blackarch' 'blackarch-recon')
pkgdesc='Python script for enumeration of hosts, subdomains and emails from a given domain using google.'
url='https://github.com/darkoperator/dnsrecon'
license=('custom')
arch=('any')
depends=('python' 'python-netaddr' 'python-dnspython' 'python-lxml'
         'python-requests' 'python-loguru')
makedepends=('git' 'python-pip' 'python-build')
source=("git+https://github.com/darkoperator/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  grep -m1 __version "$pkgname/cli.py" | cut -d '=' -f 2 | sed "s/'//g" |
  tr -d ' '
}

build() {
  cd $pkgname

  python -m build --wheel --outdir="$startdir/dist"
}

package() {
  cd $pkgname

	  pip install \
    --verbose \
    --disable-pip-version-check \
    --no-warn-script-location \
    --ignore-installed \
    --no-compile \
    --no-deps \
    --root="$pkgdir" \
    --prefix=/usr \
    --no-index \
    --find-links="file://$startdir/dist" \
    $pkgname

  rm -rf "$pkgdir/usr/lib/python$_pyver/site-packages/tests"
}

